In[1]:=
```
NotebookFileName[]
```

Out[1]=
```
C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430
  \20231204_super_glue.nb
```

Preliminaries

This is just to show how we make a string from several pieces. Below I use Epilog-> to place to line segments on the graph.

In[2]:=
```
Plot[1 + Sin[x], {x, 0, Pi},
 PlotRange → {{-1 + 0, Pi + 1}, {-2, 2}},
 Epilog →
  {{Line[{{-1, 0}, {0, 1}}], Line[{{Pi, 1}, {Pi + 1, 0}}]}},
 AspectRatio → 1/5, ImageSize → 600]
```

Out[2]=



Below I show how to “animate” the above graph using Cos[t] as a time function. The point is that I have to adjust the line segments as well.

In[3]:=
```
Manipulate[Plot[(1 + Sin[x]) Cos[t], {x, 0, Pi},
  PlotRange → {{-1 + 0, Pi + 1}, {-2, 2}},
  Epilog →
   {{Line[{{-1, 0}, {0, 1 * Cos[t]}}],
     Line[{{Pi, 1 * Cos[t]}, {Pi + 1, 0}}]}},
  AspectRatio → 1/5, ImageSize → 600], {t, 0, 2 Pi}]
```

Out[3]=



Statement of the Problem

Below is the PDE + BCs + ICs that we want to solve here.

PDE : $\partial_{tt} u[x, t] = \partial_{xx} u[x, t],\ 0 \le x \le \pi,\ t \ge 0,$

BCs : $u[0, t] = 0,\ u[\pi, t] + \partial_x u[\pi, t] = 0,$

ICs : $u[x, 0] = ff1[x],\ \partial_t u[x, 0] = 0.$

For demonstration purposes, we will look for the positive eigenvalues only and their corresponding eigenfunctions:

The general solution can be obtained using Mathematica function DSolve

Solving the Boundary-Eigenvalue Problem

Finding the eigenvalues

In[4]:=
```
DSolve[-D[A[x], {x, 2}] == μ^2 A[x], A[x], x]
```

Out[4]= $\{\{A[x] \to c_1 \,\mathrm{Cos}[x\,\mu] + c_2 \,\mathrm{Sin}[x\,\mu]\}\}$

Below I show how to substitute the special value x=0 in the general solution. Mathematica code /. is called the substitution rule.

In[5]:=
```
(c1 Cos[x μ] + c2 Sin[x μ]) /. {x → 0}
```

Out[5]= c_1

The second boundary condition is more complicated. I substitute the length of the string Pi.

In[6]:=
```
(c1 Cos[x μ] + c2 Sin[x μ] +
   D[c1 Cos[x μ] + c2 Sin[x μ], {x, 1}]) /. {x → Pi}
```

Out[6]= $c_1 \,\mathrm{Cos}[\pi\,\mu] + \mu\, c_2 \,\mathrm{Cos}[\pi\,\mu] - \mu\, c_1 \,\mathrm{Sin}[\pi\,\mu] + c_2 \,\mathrm{Sin}[\pi\,\mu]$

I use the Collect function to group terms with C[1] and C[2]. (Here Mathematica displays C[1] and C[2] in a nicer for in Output.)

In[7]:=
```
Collect[
 (ℭ1 Cos[x μ] + ℭ2 Sin[x μ] +
   D[ℭ1 Cos[x μ] + ℭ2 Sin[x μ], {x, 1}]) /. {x → Pi},
 {C[1], C[2]}]
```

Out[7]= $\mathbb{c}_2\,(\mu\,\text{Cos}[\pi\,\mu] + \text{Sin}[\pi\,\mu]) + \mathbb{c}_1\,(\text{Cos}[\pi\,\mu] - \mu\,\text{Sin}[\pi\,\mu])$

Thus, the system that we have to solve for C[1] and C[2] is:

$\mathbb{c}_1 = 0$

$\mathbb{c}_2\,(\mu\,\text{Cos}[\pi\,\mu] + \text{Sin}[\pi\,\mu]) + \mathbb{c}_1\,(\text{Cos}[\pi\,\mu] - \mu\,\text{Sin}[\pi\,\mu]) = 0$

Of course we can solve this system "by hand", but I do it slow. The determinant of the system is: (Below, I write the matrix and its determinant how Mathematica does it: A matrix is in Mathematica a list of its rows.)

In[8]:=
```
Det[
 { (* the matrix starts here *)
  {1, 0}, (* the first row, coefficients only *)
  {(Cos[π μ] - μ Sin[π μ]), (μ Cos[π μ] + Sin[π μ])}
  (* the second row, coefficients only *)
 }(* the matrix starts here *)
]
```

Out[8]= $\mu\,\text{Cos}[\pi\,\mu] + \text{Sin}[\pi\,\mu]$

Our system has a nontrivial solution for C[1] and C[2] if and only if

$\mu\,\text{Cos}[\pi\,\mu] + \text{Sin}[\pi\,\mu] = 0$

Let us see on a graph where the solutions are:

In[9]:=
```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 10}]
```

Out[9]=



The command solve cannot find the exact solutions of the equation:

```
μ Cos[π μ] + Sin[π μ] = 0
```

Try

In[10]:=
```
Solve[μ Cos[π μ] + Sin[π μ] == 0, μ]
```

Solve: This system cannot be solved with the methods available to Solve.

Out[10]=
```
Solve[μ Cos[π μ] + Sin[π μ] == 0, μ]
```

The command that will find approximate solutions is FindRoot

In[11]:=
```
FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, 0.5}]
```

Out[11]=
```
{μ → 0.787637}
```

The logic of this command is that we give an initial guess for μ and Mathematica finds a very accurate approximate value.

Below is the FindRoot as a function of the initial guess:

In[12]:=
```
(μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, #}]) &[
 0.5]
```

Out[12]= 0.787637

Verify that the value found is accurate solution:

In[13]:=
```
μ Cos[π μ] + Sin[π μ] /. {μ → 0.7876372941648639`}
```

Out[13]= 2.22045×10^{-16}

Quite close to 0.

Now try other solutions of

μ Cos[π μ] + Sin[π μ] = 0

In[14]:=
```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 10}]
```

Out[14]=


One close to 2

In[15]:=
```
(μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, #}]) &[
 2]
```

Out[15]= 1.67161

Now try more, say ten, and name the table of values:

In[16]:=
```
mymus =
 Table[(μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, #}]) &[
   s], {s, 0.5, 10.5, 1}]
```

Out[16]=
```
{0.787637, 1.67161, 2.61621, 3.58655, 4.56859,
 5.55668, 6.54824, 7.54196, 8.53712, 9.53327, 10.5301}
```

Verify on the graph if these are good solutions. The values that we calculated are at the red points.

In[17]:=
```
Map[Point[{#, 0}] &, mymus]
```

Out[17]=
```
{Point[{0.787637, 0}],
 Point[{1.67161, 0}], Point[{2.61621, 0}],
 Point[{3.58655, 0}], Point[{4.56859, 0}],
 Point[{5.55668, 0}], Point[{6.54824, 0}],
 Point[{7.54196, 0}], Point[{8.53712, 0}],
 Point[{9.53327, 0}], Point[{10.5301, 0}]}
```

In[18]:=
```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 11},
 Epilog →
  {{RGBColor[1, 0, 0], Map[Point[{#, 0}] &, mymus]}}]
```

Out[18]=



The values seem quite accurate. But we can try more of them

In[19]:=
```
Clear[mymus];
mymus =
 Table[(μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, #}]) &[
   s], {s, 0.5, 20.5, 1}]
```

Out[20]=
```
{0.787637, 1.67161, 2.61621, 3.58655, 4.56859,
 5.55668, 6.54824, 7.54196, 8.53712, 9.53327,
 10.5301, 11.5275, 12.5254, 13.5235, 14.5219, 15.5205,
 16.5192, 17.5182, 18.5172, 19.5163, 20.5155}
```

Verify on the graph if these are good solutions. The values that we calculated are at the red points.

In[21]:=
```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 21},
 Epilog →
  {{RGBColor[1, 0, 0], Map[Point[{#, 0}] &, mymus]}}]
```

Out[21]=



Let us verify the values numerically as well. Here we use Map[] function

```
(# Cos[π #] + Sin[π #]) &
```

is so called Pure Function, instead of the variable we put # and finish the Pure Function with &

In[22]:=
```
Map[(# Cos[π #] + Sin[π #]) &, mymus]
```

Out[22]= $\{2.22045\times10^{-16}, 1.09912\times10^{-14}, -1.22125\times10^{-15}, 6.66134\times10^{-16}, -1.11022\times10^{-15}, 1.13243\times10^{-14}, -3.57492\times10^{-14}, 1.31006\times10^{-14}, 1.05471\times10^{-14}, 1.72085\times10^{-14}, 4.94049\times10^{-14}, -7.63833\times10^{-14}, -1.88738\times10^{-14}, -2.64233\times10^{-14}, -5.73985\times10^{-14}, -6.33937\times10^{-14}, 4.996\times10^{-14}, -2.87548\times10^{-14}, 1.9984\times10^{-15}, 1.07914\times10^{-13}, -1.24456\times10^{-13}\}$

The above is quite impressive. We can make it even more impressive by Chop[]

```
In[23]:= Chop[Map[(# Cos[π #] + Sin[π #]) &, mymus]]

Out[23]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Finding the eigenfunctions

Thus, we have 21 values for μ-s which are in a list which we call **mymus;** the squares of μ-s are the eigenvalues. Now write the corresponding eigenfunctions. For the μ-s that we found we need a nontrivial solution of the system

$$\mathbb{C}_1 = 0$$

$$\mathbb{C}_2 (\mu \operatorname{Cos}[\pi \mu] + \operatorname{Sin}[\pi \mu]) + \mathbb{C}_1 (\operatorname{Cos}[\pi \mu] - \mu \operatorname{Sin}[\pi \mu]) = 0$$

A nontrivial solution is C[1]=0 and C[2]=1. Thus the corresponding eigenfunctions are: (Go back to the general solution and plug in C[1]=0 and C[2]=1 and use mus in the list mymus:

```
In[24]:= (ℂ1 Cos[x μ] + ℂ2 Sin[x μ]) /. {C[1] → 0, C[2] → 1}

Out[24]= Sin[x μ]

In[25]:= Clear[myeigfuns];
         myeigfuns[x_] = Sin[x μ] /. {μ → mymus}

Out[25]= {Sin[0.787637 x], Sin[1.67161 x], Sin[2.61621 x],
          Sin[3.58655 x], Sin[4.56859 x], Sin[5.55668 x],
          Sin[6.54824 x], Sin[7.54196 x], Sin[8.53712 x],
          Sin[9.53327 x], Sin[10.5301 x], Sin[11.5275 x],
          Sin[12.5254 x], Sin[13.5235 x], Sin[14.5219 x],
          Sin[15.5205 x], Sin[16.5192 x], Sin[17.5182 x],
          Sin[18.5172 x], Sin[19.5163 x], Sin[20.5155 x]}
```

The eigenfunctions are in the list called **myeigfuns[x]**. The

Mathematica code for selecting particular elements of a list is as follows, the second element

In[26]:=
```
myeigfuns[x][[2]]
```

Out[26]=
```
Sin[1.67161 x]
```

These are our eigenfunctions. I made a claim that they are orthogonal to each other. Let us verify that, just say the first two

In[27]:=
```
myeigfuns[x][[1]]
```

Out[27]=
```
Sin[0.787637 x]
```

In[28]:=
```
myeigfuns[x][[2]]
```

Out[28]=
```
Sin[1.67161 x]
```

Show that they are orthogonal:

In[29]:=
```
Integrate[myeigfuns[x][[1]] × myeigfuns[x][[2]],
  {x, 0, Pi}]
```

Out[29]= -3.2474×10^{-15}

Now try the entire table of all integrals. That will be 21x21 matrix which should have only diagonal elements nonzero: (The command below is slow. Using NIntegrate, numerical integration would be faster. But, NIntegrate might require special options to be successful. You can find the options in the bigger notebook that I posted on the class website.)

In[30]:=
```
21 × 21
```

Out[30]=
```
441
```

In[31]:=
```
Table[
  Chop[Integrate[myeigfuns[x][[j]] × myeigfuns[x][[k]],
    {x, 0, Pi}]], {j, 1, 21}, {k, 1, 21}]
```

Out[31]=
```
{{1.87937, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 1.70257, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 1.63453, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 1.60686, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 1.59366, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 1.58648, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0,
  1.58219, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 1.57943, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0,
  1.57756, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57624, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 1.57527, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57453, 0, 0, 0, 0,
  0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  1.57396, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 1.57352, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57316, 0,
  0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 1.57286, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57262, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  1.57242, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57225, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 1.57211, 0}, {0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.57198}}
```

Constructing natural modes of vibrations

Next is to do time part. That is similar. Time part has two functions Cos and Sin. Let us ignore Sin, which corresponds to the initial velocity. The time Cos parts are

In[32]:=
```
Cos[t μ] /. {μ → mymus}
```

Out[32]=
```
{Cos[0.787637 t], Cos[1.67161 t], Cos[2.61621 t],
 Cos[3.58655 t], Cos[4.56859 t], Cos[5.55668 t],
 Cos[6.54824 t], Cos[7.54196 t], Cos[8.53712 t],
 Cos[9.53327 t], Cos[10.5301 t], Cos[11.5275 t],
 Cos[12.5254 t], Cos[13.5235 t], Cos[14.5219 t],
 Cos[15.5205 t], Cos[16.5192 t], Cos[17.5182 t],
 Cos[18.5172 t], Cos[19.5163 t], Cos[20.5155 t]}
```

In[33]:=
```
Clear[mytimeCs];
mytimeCs[t_] = Cos[t μ] /. {μ → mymus}
```

Out[33]=
```
{Cos[0.787637 t], Cos[1.67161 t], Cos[2.61621 t],
 Cos[3.58655 t], Cos[4.56859 t], Cos[5.55668 t],
 Cos[6.54824 t], Cos[7.54196 t], Cos[8.53712 t],
 Cos[9.53327 t], Cos[10.5301 t], Cos[11.5275 t],
 Cos[12.5254 t], Cos[13.5235 t], Cos[14.5219 t],
 Cos[15.5205 t], Cos[16.5192 t], Cos[17.5182 t],
 Cos[18.5172 t], Cos[19.5163 t], Cos[20.5155 t]}
```

In[34]:=
```
mytimeCs[t][[1]]
```

Out[34]=
```
Cos[0.787637 t]
```

Now we have found the natural modes of vibrations of this string: The first natural mode of vibration is

In[35]:=
```
myeigfuns[x][[1]] × mytimeCs[t][[1]]
```

Out[35]=
```
Cos[0.787637 t] Sin[0.787637 x]
```

See the value at at t=0

In[36]:=
```
myeigfuns[x]⟦1⟧ (mytimeCs[0]⟦1⟧)
```

Out[36]=
```
1. Sin[0.787637 x]
```

In[37]:=
```
Plot[myeigfuns[x]⟦1⟧ × mytimeCs[0]⟦1⟧, {x, 0, Pi},
 PlotRange → {{0, Pi + 1}, {-1, 1}},
 Epilog →
  {{Line[{{Pi, myeigfuns[Pi][[1]] × mytimeCs[0][[1]]},
      {Pi + 1, 0}}]}}, AspectRatio → 1/5, ImageSize → 600]
```

Out[37]=



Let us animate it:

In[38]:=
```
Manipulate[Plot[myeigfuns[x][[1]] × mytimeCs[t][[1]],
  {x, 0, Pi}, PlotRange → {{0, Pi + 1}, {-1, 1}},
  Epilog →
   {{Line[{{Pi, myeigfuns[Pi][[1]] × mytimeCs[t][[1]]},
       {Pi + 1, 0}}]}}, AspectRatio → 1/5, ImageSize → 600],
 {t, 0, 2 Pi/mymus[[1]]}]
```

Out[38]=



In[39]:=
```
Manipulate[Plot[myeigfuns[x][[2]] × mytimeCs[t][[2]],
  {x, 0, Pi}, PlotRange → {{0, Pi + 1}, {-1.1, 1.1}},
  Epilog →
   {{Line[{{Pi, myeigfuns[Pi][[2]] × mytimeCs[t][[2]]},
        {Pi + 1, 0}}]}}, AspectRatio → 1/5, ImageSize → 600],
 {t, 0, 2 Pi/mymus[[2]]}]
```

Out[39]=



In[40]:=
```
Manipulate[Plot[myeigfuns[x][[k]] × mytimeCs[t][[k]],
   {x, 0, Pi}, PlotRange → {{0, Pi + 1}, {-1.1, 1.1}},
   Epilog →
    {{Line[{{Pi, myeigfuns[Pi][[k]] × mytimeCs[t][[k]]},
        {Pi + 1, 0}}]}}, AspectRatio → 1/5, ImageSize → 600],
  {t, 0, 2 Pi/mymus[[k]], Appearance → {"Open"}},
  {k, Range[21], Setter, Appearance → {"Open"}}]
```

Out[40]=



Solving the Initial Value Problem

Now introduce a specific initial condition. It must satisfy the boundary conditions. Let us try with a quadratic function $x^2 + b\,x$ and choose b to satisfy the second boundary condition:

In[41]:= ((x^2 + b x) /. {x → Pi}) + (D[(x^2 + b x), x]) /. {x → Pi}

Out[41]= $b + 2\pi + b\pi + \pi^2$

In[42]:= Solve[
(((x^2 + b x) /. {x → Pi}) + (D[(x^2 + b x), x]) /. {x → Pi}) == 0,
b]

Out[42]= $\left\{\left\{b \to \frac{-2\pi - \pi^2}{1 + \pi}\right\}\right\}$

Try the following initial condition

In[43]:= (x^2 + b x) /. $\left\{b \text{ -> } \frac{-2\pi - \pi^2}{1 + \pi}\right\}$

Out[43]= $\frac{(-2\pi - \pi^2)\, x}{1 + \pi} + x^2$

In[44]:= Clear[ff]; ff[x_] = $\frac{1}{4}\left(\frac{(-2\pi - \pi^2)\, x}{1 + \pi} + x^2\right)$

Out[44]= $\frac{1}{4}\left(\frac{(-2\pi - \pi^2)\, x}{1 + \pi} + x^2\right)$

In[45]:=
```
Plot[ff[x], {x, 0, Pi}, PlotRange → {{0, Pi + 1}, {-1, 1}},
  Epilog → {{Line[{{Pi, ff[Pi]}, {Pi + 1, 0}}]}},
  AspectRatio → 1/5, ImageSize → 600]
```

Out[45]=


In[46]:=
```
myeigfuns[x][[4]]
```

Out[46]=
```
Sin[3.58655 x]
```

In[47]:=
```
Integrate[ff[x] × myeigfuns[x][[1]], {x, 0, Pi}] /
  Integrate[myeigfuns[x][[1]] × myeigfuns[x][[1]],
   {x, 0, Pi}]
```

Out[47]=
```
-0.972209
```

In[48]:=
```
Clear[coeffs];
coeffs =
  Table[Integrate[ff[x] × myeigfuns[x][[k]], {x, 0, Pi}] /
    Integrate[myeigfuns[x][[k]] × myeigfuns[x][[k]],
     {x, 0, Pi}], {k, 1, 21}]
```

Out[48]=
```
{-0.972209, -0.0305953, -0.0231819, -0.00493321,
 -0.00399378, -0.00151156, -0.00129539, -0.000640936,
 -0.000568651, -0.000327925, -0.00029754, -0.000189388,
 -0.000174527, -0.000119004, -0.000110914,
 -0.0000795612, -0.000074792, -0.0000557767,
 -0.0000527879, -0.000040596, -0.0000386297}
```

In[49]:=
```
Clear[appff];
appff[x_] = Sum[coeffs[[k]] × myeigfuns[x][[k]], {k, 1, 21}]
```

Out[49]=
```
-0.972209 Sin[0.787637 x] -
 0.0305953 Sin[1.67161 x] - 0.0231819 Sin[2.61621 x] -
 0.00493321 Sin[3.58655 x] - 0.00399378 Sin[4.56859 x] -
 0.00151156 Sin[5.55668 x] - 0.00129539 Sin[6.54824 x] -
 0.000640936 Sin[7.54196 x] - 0.000568651 Sin[8.53712 x] -
 0.000327925 Sin[9.53327 x] - 0.00029754 Sin[10.5301 x] -
 0.000189388 Sin[11.5275 x] - 0.000174527 Sin[12.5254 x] -
 0.000119004 Sin[13.5235 x] - 0.000110914 Sin[14.5219 x] -
 0.0000795612 Sin[15.5205 x] - 0.000074792 Sin[16.5192 x] -
 0.0000557767 Sin[17.5182 x] - 0.0000527879 Sin[18.5172 x] -
 0.000040596 Sin[19.5163 x] - 0.0000386297 Sin[20.5155 x]
```

In[50]:=
```
Plot[{ff[x], appff[x]}, {x, 0, Pi},
 PlotRange → {{0, Pi}, {-1, 1}}, AspectRatio → 1/5,
 ImageSize → 600]
```

Out[50]=



An approximation for the solution is:

In[51]:=
```
Clear[appuu];
appuu[x_, t_] =
 Sum[coeffs[[k]] × myeigfuns[x][[k]] × mytimeCs[t][[k]],
  {k, 1, 21}]
```

Out[51]=
```
-0.972209 Cos[0.787637 t] Sin[0.787637 x] -
 0.0305953 Cos[1.67161 t] Sin[1.67161 x] -
 0.0231819 Cos[2.61621 t] Sin[2.61621 x] -
 0.00493321 Cos[3.58655 t] Sin[3.58655 x] -
 0.00399378 Cos[4.56859 t] Sin[4.56859 x] -
 0.00151156 Cos[5.55668 t] Sin[5.55668 x] -
 0.00129539 Cos[6.54824 t] Sin[6.54824 x] -
 0.000640936 Cos[7.54196 t] Sin[7.54196 x] -
 0.000568651 Cos[8.53712 t] Sin[8.53712 x] -
 0.000327925 Cos[9.53327 t] Sin[9.53327 x] -
 0.00029754 Cos[10.5301 t] Sin[10.5301 x] -
 0.000189388 Cos[11.5275 t] Sin[11.5275 x] -
 0.000174527 Cos[12.5254 t] Sin[12.5254 x] -
 0.000119004 Cos[13.5235 t] Sin[13.5235 x] -
 0.000110914 Cos[14.5219 t] Sin[14.5219 x] -
 0.0000795612 Cos[15.5205 t] Sin[15.5205 x] -
 0.000074792 Cos[16.5192 t] Sin[16.5192 x] -
 0.0000557767 Cos[17.5182 t] Sin[17.5182 x] -
 0.0000527879 Cos[18.5172 t] Sin[18.5172 x] -
 0.000040596 Cos[19.5163 t] Sin[19.5163 x] -
 0.0000386297 Cos[20.5155 t] Sin[20.5155 x]
```

Test it with specific values:

In[52]:=
```
appuu[1, 1]
```

Out[52]=
```
-0.475038
```

Animate

In[53]:=
```
Manipulate[Plot[appuu[x, t], {x, 0, Pi},
   PlotRange → {{0, Pi + 1}, {-1, 1}},
   Epilog → {{Line[{{Pi, appuu[Pi, t]}, {Pi + 1, 0}}]}},
   AspectRatio → 1/5, ImageSize → 600],
  {t, 0, 8, Appearance → {"Open"}}]
```

Out[53]=



Solving the Initial Value Problem with a more Complicated $f(x)$

In this section I introduce a more complicated function $f(x)$ for which symbolic Integrate[] can not calculate the coefficients. We use NIntegrate[] with specific options.

I just multiply the preceding initial displacement with a high power of the Sin[] function:

In[54]:=
```
Clear[ff1]; ff1[x_] = 1/4 (((-2 π - π^2) x)/(1 + π) + x^2) Sin[x]^6
```

Out[54]= $\frac{1}{4}\left(\frac{\left(-2\pi-\pi^2\right)x}{1+\pi}+x^2\right)\text{Sin}[x]^6$

Satisfies the boundary conditions:

In[55]:=
```
ff1[x] /. {x → 0}
```

Out[55]= 0

In[56]:=
```
(ff1[x] /. {x → Pi}) - D[ff1[x], {x, 1}] /. {x → Pi}
```

Out[56]= 0

See the function

In[57]:=
```
Plot[ff1[x], {x, 0, Pi}, PlotRange → {{0, Pi + 1}, {-1, 1}},
 Epilog → {{Line[{{Pi, ff[Pi]}, {Pi + 1, 0}}]}},
 AspectRatio → 1/5, ImageSize → 600]
```

Out[57]=



In[58]:=
```
myeigfuns[x][[4]]
```

Out[58]= Sin[3.58655 x]

See how Integrate[] is slower than NIntegrate[]

In[59]:=
```
tI =
 Timing[Integrate[ff1[x] × myeigfuns[x][[1]], {x, 0, Pi}] /
   Integrate[myeigfuns[x][[1]] × myeigfuns[x][[1]],
    {x, 0, Pi}]]
```

Out[59]=
```
{3.32813, -0.420023}
```

Pay attention how we need special options in NIntegrate[]

In[60]:=
```
tNI =
 Timing[
  NIntegrate[ff1[x] × myeigfuns[x][[1]], {x, 0, Pi},
    MaxRecursion → 200, AccuracyGoal → 10,
    PrecisionGoal → 10] /
   NIntegrate[myeigfuns[x][[1]] × myeigfuns[x][[1]],
    {x, 0, Pi}, MaxRecursion → 200, AccuracyGoal → 10,
    PrecisionGoal → 10]]
```

Out[60]=
```
{0.015625, -0.420023}
```

NIntegrate is

In[61]:=
```
tI[[1]] / tNI[[1]]
```

Out[61]=
```
213.
```

times faster.

Below I calculate the coefficients for the initial coefficient **ff1[x]**

In[62]:=
```
Clear[coeff1s];
coeff1s =
  Table[NIntegrate[ff1[x] × myeigfuns[x][[k]], {x, 0, Pi},
      MaxRecursion → 200, AccuracyGoal → 10,
      PrecisionGoal → 10] /
    NIntegrate[myeigfuns[x][[k]] × myeigfuns[x][[k]],
     {x, 0, Pi}, MaxRecursion → 200, AccuracyGoal → 10,
     PrecisionGoal → 10], {k, 1, 21}]
```

Out[62]= $\{-0.420023, -0.188822, 0.289626, 0.114478, -0.108061, -0.0283802, 0.0175115, 0.00191776, -0.0000629462, 0.0000958602, -0.0000121313, 0.0000134631, -2.83361\times10^{-6}, 2.95606\times10^{-6}, -8.30507\times10^{-7}, 8.47274\times10^{-7}, -2.88305\times10^{-7}, 2.91139\times10^{-7}, -1.13753\times10^{-7}, 1.14263\times10^{-7}, -4.9598\times10^{-8}\}$

Here is our approximation of the initial condition **ff1[x]**

In[63]:=
```
Clear[appff1];
appff1[x_] = Sum[coeff1s[[k]] × myeigfuns[x][[k]],
   {k, 1, 21}]
```

Out[63]= $-0.420023\,\mathrm{Sin}[0.787637\,x] - 0.188822\,\mathrm{Sin}[1.67161\,x] + 0.289626\,\mathrm{Sin}[2.61621\,x] + 0.114478\,\mathrm{Sin}[3.58655\,x] - 0.108061\,\mathrm{Sin}[4.56859\,x] - 0.0283802\,\mathrm{Sin}[5.55668\,x] + 0.0175115\,\mathrm{Sin}[6.54824\,x] + 0.00191776\,\mathrm{Sin}[7.54196\,x] - 0.0000629462\,\mathrm{Sin}[8.53712\,x] + 0.0000958602\,\mathrm{Sin}[9.53327\,x] - 0.0000121313\,\mathrm{Sin}[10.5301\,x] + 0.0000134631\,\mathrm{Sin}[11.5275\,x] - 2.83361\times10^{-6}\,\mathrm{Sin}[12.5254\,x] + 2.95606\times10^{-6}\,\mathrm{Sin}[13.5235\,x] - 8.30507\times10^{-7}\,\mathrm{Sin}[14.5219\,x] + 8.47274\times10^{-7}\,\mathrm{Sin}[15.5205\,x] - 2.88305\times10^{-7}\,\mathrm{Sin}[16.5192\,x] + 2.91139\times10^{-7}\,\mathrm{Sin}[17.5182\,x] - 1.13753\times10^{-7}\,\mathrm{Sin}[18.5172\,x] + 1.14263\times10^{-7}\,\mathrm{Sin}[19.5163\,x] - 4.9598\times10^{-8}\,\mathrm{Sin}[20.5155\,x]$

How good is it?

In[64]:=
```
Plot[{ff1[x], appff1[x]}, {x, 0, Pi},
 PlotStyle → {{RGBColor[0, 0, 0.7], Thickness[0.01]},
   {RGBColor[1, 0, 0], Thickness[0.004]}},
 PlotRange → {{0, Pi}, {-1, 1}}, AspectRatio → 1/5,
 ImageSize → 600]
```

Out[64]=



Quite good approximation!

An approximation for the solution of PDE+BCs+ICs:

In[65]:=
```
Clear[appuu1];
appuu1[x_, t_] =
 Sum[coeff1s[[k]] × myeigfuns[x][[k]] × mytimeCs[t][[k]],
  {k, 1, 21}]
```

Out[65]=
```
-0.420023 Cos[0.787637 t] Sin[0.787637 x] -
 0.188822 Cos[1.67161 t] Sin[1.67161 x] +
 0.289626 Cos[2.61621 t] Sin[2.61621 x] +
 0.114478 Cos[3.58655 t] Sin[3.58655 x] -
 0.108061 Cos[4.56859 t] Sin[4.56859 x] -
 0.0283802 Cos[5.55668 t] Sin[5.55668 x] +
 0.0175115 Cos[6.54824 t] Sin[6.54824 x] +
 0.00191776 Cos[7.54196 t] Sin[7.54196 x] -
 0.0000629462 Cos[8.53712 t] Sin[8.53712 x] +
 0.0000958602 Cos[9.53327 t] Sin[9.53327 x] -
 0.0000121313 Cos[10.5301 t] Sin[10.5301 x] +
 0.0000134631 Cos[11.5275 t] Sin[11.5275 x] -
 2.83361 × 10^-6 Cos[12.5254 t] Sin[12.5254 x] +
 2.95606 × 10^-6 Cos[13.5235 t] Sin[13.5235 x] -
 8.30507 × 10^-7 Cos[14.5219 t] Sin[14.5219 x] +
 8.47274 × 10^-7 Cos[15.5205 t] Sin[15.5205 x] -
 2.88305 × 10^-7 Cos[16.5192 t] Sin[16.5192 x] +
 2.91139 × 10^-7 Cos[17.5182 t] Sin[17.5182 x] -
 1.13753 × 10^-7 Cos[18.5172 t] Sin[18.5172 x] +
 1.14263 × 10^-7 Cos[19.5163 t] Sin[19.5163 x] -
 4.9598 × 10^-8 Cos[20.5155 t] Sin[20.5155 x]
```

Test it with specific values:

In[66]:=
```
appuu1[1, 1]
```

Out[66]=
```
-0.26851
```

Animate

In[67]:=
```
Manipulate[Plot[appuu1[x, t], {x, 0, Pi},
   PlotRange → {{0, Pi + 1}, {-1, 1}},
   Epilog → {{Line[{{Pi, appuu1[Pi, t]}, {Pi + 1, 0}}]}},
   AspectRatio → 1/5, ImageSize → 600],
  {t, 0, 12.8, Appearance → {"Open"}}]
```

Out[67]=